


FORTRAN II Program for Standard-Size Analysis of Unconsolidated Sediments Using an IBM 1620 Computer

By
Jack W. Pierce
 U.S. National Museum and
 George Washington University
 and
Donald I. Good

SPECIAL DISTRIBUTION PUBLICATION 28



State Geological Survey
 The University of Kansas, Lawrence
 1966

EDITORIAL STAFF

Daniel F. Merriam, Editor

Associate Editors

John R. Dempsey
 Richard W. Fetzner
 James M. Forgotsen, Jr.
 John C. Griffiths
 John W. Harbaugh
 Richard G. Hetherington
 Sidney N. Hockens
 John Imbrie
 William C. Pearn
 Max G. Pitcher
 Floyd W. Preston
 Peter H.A. Sneath
 Owen T. Spitz

Northern Natural Gas Co.
 Sun Oil Company
 Pan American Petroleum Corp.
 Pennsylvania State University
 Stanford University
 University of Kansas
 Phillips Petroleum Co.
 Columbia University
 Socony Mobil Field Research Lab.
 Continental Oil Co.
 University of Kansas
 University of Leicester
 Kansas Geological Survey

Price of punched card program and data decks that are available for limited time. Address all correspondence to Editor, Computer Contributions, Kansas Geological Survey, The University of Kansas, Lawrence, Kansas.

	BALGOL	ALGOL	FORTRAN II	FORTRAN IV
Trend-3 (SDP 3)	\$10.00			
Match-Coeff (SDP 4)			\$2.00	
Correlation and distance Coeff (SDP 9)	\$ 5.00 (each)			
Time-trend (SDP 12)			\$5.00	\$5.00
Covap (SDP 13)			\$15.00	
Trend-3 (SDP 14)			\$25.00	\$25.00
Cross-Association (SDP 23)				\$10.00
Single and double Fourier (SDP 24)	\$ 5.00 \$15.00			
Precambrian wells (SDP 25)	List of about 2,600 Precambrian wells	\$50.00		
Trend-4 (SDP 26)			\$7.50	
Sediment analysis (SDP 28)			\$10.00	
4D Trend (KGS B171)	\$10.00			
Conversion of T&R to Cartesian coordinates (Bull 170-3)			\$5.00	

SEE INSIDE BACK COVER FOR LIST OF AVAILABLE COMPUTER CONTRIBUTIONS

FORTRAN II PROGRAM FOR STANDARD-SIZE ANALYSIS OF UNCONSOLIDATED SEDIMENTS

Using an IBM 1620 Computer

by

Jack W. Pierce and Donald I. Good

INTRODUCTION

This program computes the mean, standard deviation, skewness, and kurtosis by the method of moments for size distribution of particles as determined by standard sediment analysis. Specifically, it analyzes Recent sediments. Both sieve and pipette analyses were performed.

The method of moments, although the most concise method for determination of the mean, standard deviation, skewness, and kurtosis, has been little used because of computational time involved. Trask (1932), Inman (1952), and Mason and Folk (1958) proposed the use of graphical methods to describe the size distribution or to obtain approximations of the parameters. The method of moments is limited in that distribution must not be "open-ended," that is, only a very minor portion of the material can be outside the range of the analytical technique employed. This might occur only when a significant portion of the sample occurs in the finer part of the clay-size range.

The program gives information about the sediment in addition to the four size parameters. Part of the output is the fraction of the total sample in each size class, and allows the user to plot a cumulative distribution curve or a histogram.

According to Shepard and Moore (1954), the percent of one or more "coarse-fraction" components in the total sample is indicative of the environment of deposition. The method of "coarse-fraction" analysis is discussed by Shepard (1963, p. 149). This program calculates the fraction of total sediment, in the larger-than-silt-size fraction, made up of ten different components.

Part of this program computes the size parameters for the distribution of minerals resistant to weathering in the sediment. Some components of the sediment, such as shell fragments and plant material, did not undergo extensive transport before deposition. Therefore, it is desired to examine the distribution of the transported material.

Although the program was written for a specific geologic problem, no modifications would be necessary for a general size-analysis problem. Slight modifications would allow it to be used for any problem computing the mean and associated parameters by the method of moments.

Acknowledgments. - This program was developed using the facilities at the University of Kansas Computation Center. We thank Dr. Floyd W. Preston and Dean A. Lebestky for assistance in developing the program.

MATHEMATICAL DESCRIPTION

The most accurate method to obtain the mean, standard deviation, skewness, and kurtosis for any series of grouped measurements, is the method of moments. This program is based upon the moment method for calculating the statistical parameters.

The weighted arithmetic mean, \bar{x} , of a series of related measurements, is defined as

$$\bar{x} = \frac{1}{n} \sum_{i=1}^k f(x_i)x_i \quad (1)$$

For sediment-size analyses, n is the total weight of the samples; $f(x_i)$ is the weight of a single size class; x_i is the midpoint of the same size class in phi scale units; and k is the total number of size classes.

The sample standard deviation, s , taken as the square root of the sample variance, is

$$s = \left[\frac{1}{n} \sum_{i=1}^k f(x_i) (x_i - \bar{x})^2 \right]^{1/2} \quad (2)$$

Skewness and kurtosis are associated with the third and fourth moments about the mean, but have been defined by different authors in several ways. For the purposes of this program, skewness, Sk , is defined as

$$Sk = \frac{1}{2n} \sum_{i=1}^k \frac{f(x_i) (x_i - \bar{x})^3}{s^3} \quad (3)$$

and kurtosis, Ku , as

$$Ku = \frac{1}{2} \left[\sum_{i=1}^k \frac{f(x_i) (x_i - \bar{x})^4}{ns^4} - 3 \right] \quad (4)$$

The fraction, y_i , of the total sample in a single size class, $f(x_i)$, is given by

$$y_i = \frac{f(x_i)}{n} \quad (5)$$

The fraction, z_i , of any coarse-fraction component in the total sample is given by

$$z_i = \frac{\sum_{j=1}^k p_{ij} f(x_j)}{n} \quad , \quad (6)$$

where p_{ij} is the fraction of the constituent in a single class, $f(x_j)$ is the weight of this size class, and n is the total weight of the sample.

LISTING OF NOMENCLATURE

AB	PMPHI (J) - XMEAN.
AMPHI (I)	Midpoint of a single sieve-size class (in phi units).
A1	Weight of shell fragments in a single sieve-size class.
A2	Weight of organic material in a single sieve-size class.
A3	Weight of clay aggregates in a single sieve-size class.
A4	Weight of ferruginous aggregates in a single sieve-size class.
A5	Weight of plant fragments in a single sieve-size class.
A6	Weight of mud (silt and clay) in a single sieve-size class.
A7	Weight of echinoid plates and spines in a single sieve-size class.
BEAK	Weight of the container in which the total sample was weighed.
BEAKR (I)	Weight of the container in which a sieve fraction was weighed.
CAGG (I)	Estimated fraction of clay aggregates in a single sieve-size class.
CLPCT	Fraction of material in clay-size range.
CURT	Kurtosis, K_u .
DIV	Reciprocal of the sum of the net weight of all size classes, $1./SUMFR$.
ECH (I)	Estimated fraction of echinoid plates and spines in a single sieve-size class.
FAGG (I)	Estimated fraction of ferruginous aggregates in a single sieve-size class.
FMUD (I)	Estimated fraction of mud (silt and clay) in a single sieve-size class.
FPPCT (J)	Fraction of material in a single pipette interval.
FRGRS (I)	Gross weight of the sediment in one sieve interval.
FRNT (I)	Net weight of a single sieve-size class, $FRGRS (I) - BEAKR (I)$.
FRPCT (I)	Fraction of material in a single sieve-size class.
GROSS	Weight of the total sample.
I	Reference for the sieve-size range.
IM	The number of sieve classes less one.
J	Reference for the pipette-size range.
K	A counter that ties the data output to the sample number. It increases by one each time new sample data is read into the program.
L	Index for calculating weight of each pipette fraction.
M	Number of sieve-size classes in a particular sample.
MM	$MM = M$ if pipettes are absent; $MM = M - 1 = IM$ if pipettes are present.
N	Number of samples for which there is data.
OPAQ (I)	Estimated fraction of opaque minerals in a single sieve-size class.
ORG (I)	Estimated fraction of organic material in a single sieve-size class.
P	Indicator that shows the presence or absence of a pipette analysis. Zero (0) indicates pipette data absent; 1, pipette data present.
PCAGG	Fraction of clay aggregates in the total sample.
PECH	Fraction of echinoid plates and spines in the total sample.
PFAGG	Fraction of ferruginous aggregates in the total sample.
PFMUD	Fraction of mud (silt and clay) in the total sample.
PIP (J)	Measured net weight of sediment in a single pipette fraction.
PLFRG (I)	Estimated fraction of plant fragments in a single sieve-size class.
PMPHI (J)	Midpoint of a single pipette fraction (in phi units).
POPAQ	Fraction of opaque minerals in the total sample.
PORG	Fraction of organic material in the total sample.

PPLFRG	Fraction of plant fragments in the total sample.
PQTZ	Fraction of quartz in the total sample.
PSHELL	Fraction of shell fragments in the total sample.
PYQTZ	Fraction of yellow quartz in the total sample.
QTZ (I)	Estimated fraction of quartz in a single sieve-size class.
SCAGG	Sum of the weight of clay aggregates in all size fractions.
SDPCT	Fraction of material in sand-size range.
SECH	Sum of the weight of echinoid plates and spines in all size fractions.
SFAGG	Sum of the weight of ferruginous aggregates in all size fractions.
SFMUD	Sum of the weight of mud (silt and clay) in all size fractions.
SHELL (I)	Estimated fraction of shell fragments in a single sieve-size class.
SIGMA	Standard deviation, s .
SIGMA3	$(\text{SIGMA})^3$, cube of standard deviation.
SIGMA4	$(\text{SIGMA})^4$, standard deviation raised to fourth power.
SKEW	Skewness, Sk .
SLOSS	Difference between sample net weight and sum of the fraction net weights.
SLPCT	Fraction of material in silt-size range.
SMPNT	Net weight of the sample, $\text{GROSS} - \text{BEAK}$.
SOPAQ	Sum of the weight of opaque minerals in all size fractions.
SORG	Sum of the weight of organic material in all size fractions.
SPLFRG	Sum of the weight of plant fragments in all size fractions.
SQTZ	Sum of the weight of quartz in all size fractions.
SSHELL	Sum of the weight of shell fragments in all size fractions.
SUMFR	Sum of the net weights of all sieve-size classes.
SUMFR1	Sum of the net weight of sieve-size classes with the sum of pipette weights substituted for the net weight of the pan fraction.
SUMWP	Sum of the net pipette weights.
SUMX	$\text{SUMX1} + \text{SUMX2}$.
SUMX1	Sum of weights of size classes multiplied by the midpoint of the respective sieve-size class.
SUMX2	Sum of weights of size fraction multiplied by the midpoint of the respective pipette-size fraction.
SUM2A	$\sum_{i=1}^M f(x_i) (x_i - \bar{x})^2 \quad .$
SUM2B	Intermediate step for standard deviation; applicable to sieve analysis.
SUM3A	Same as SUM2A except in the pipette-size range.
	$\sum_{i=1}^M f(x_i) (x_i - \bar{x})^3 \quad .$
SUM3B	Intermediate step for skewness; applicable to sieve analysis.
SUM4A	Same as SUM3A except in the pipette-size range.
	$\sum_{i=1}^M f(x_i) (x_i - \bar{x})^4 \quad .$
SUM4B	Intermediate step for kurtosis; applicable to sieve analysis.
SYQTZ	Same as SUM4A except in the pipette-size range.
WTPIP (J)	Sum of the weight of yellow quartz in all size fractions.
XMEAN	Calculated total weight of material in a pipette fraction.
YQTZ (I)	The mean of the size distribution, \bar{x} .
	Estimated fraction of yellow quartz in a single sieve-size class.

RELATION OF LABORATORY TECHNIQUES AND PROGRAM

Standard sediment-analysis techniques were employed to generate data input for this program. Details of these techniques are discussed in Krumbein and Pettijohn (1938) and Milner (1962).

The sample and associated weighing vessels were weighed to the nearest milligram. Sand and larger size material was separated into 1/4 phi- or 1/2 phi-size fractions by sieve analysis. The program is designed to accommodate any number of classes up to 22. The material in each size class and associated weighing vessel was weighed to the nearest milligram.

For material finer than sand size, pipette analysis was used to determine the amount of material in the following size ranges (in phi-scale units): 4.0 to 4.5, 4.5 to 5.0, 5.0 to 5.5, 5.5 to 6.0, 6.0 to 7.0, 7.0 to 8.0, 8.0 to 9.0, and less than 9 phi. Each 20-milliliter aliquot portion was weighed to 0.1 milligram. Wet sieving was used to separate the sand and larger size material from the mud before sieve or pipette analyses were run.

No material remains in the pan, after sieve analysis if a satisfactory separation of the fines is accomplished by wet sieving. Although the weight on the pan fraction is zero, it must be entered in the data input as 0.0000. Unless this is done, the sum of the pipette weights will be substituted for the next larger size class and erroneous answers will result.

Sodium hexametaphosphate, with a concentration of 0.33 grams per liter, was used as a dispersant. This concentration gave 0.0066 grams of dispersant per 20-milliliter aliquot portion. If a different concentration of dispersant is used, the appropriate value must be substituted for 0.0066 in program statement 174.

Sand and larger size fractions were recombined into 1 phi classes after sieving. These were examined under a binocular microscope and the percentage of 10 different components was estimated for each size class. Type of components selected is immaterial; types significant to the study are used. If less than 10 components are used, the extra places are left blank. In this study, two types of quartz, opaque minerals, shell fragments, decaying organic matter, clay aggregates, ferruginous aggregates, plant fragments, echinoid plates, and mud were used.

Although estimates were made on the recombined 1-phi-size classes, the fractional figure must be given for each size class for which there is data. It was assumed that the estimates for the material in a 1-phi-size class applied equally well to all parts of this class before recombining.

MAJOR STEPS IN THE SIZE-ANALYSIS PROGRAM

Major steps in the program are shown on the flow chart (Fig. 1) and are described below. Statement numbers refer to those in the detailed program (Table 1).




Figure 1. - Generalized flow chart for standard-size analysis program.

- (1) Read number of samples for which there is data, statement 20.
- (2) Zero input variables, statement 35 through 63.
- (3) Read input data for sample, statements 65 through 100.
- (4) Calculate and punch net sample weight and difference between gross sample weight and sum of size fractions or analytical loss, statements 105 through 165.
- (5) If control card indicates data from a pipette analysis, continue to (6); if no pipette data, go to (8); statement 170.
- (6) Calculate fraction of total sample in each pipette-size class, statements 174 through 205.
- (7) Substitute sum of pipette weights for weight of pan fraction, which should be zero; statements 225 through 246.
- (8) Calculate and punch fraction of total sample in each size class; statements 250 through 290.
- (9) Calculate and punch fraction of total sample in sand size, silt size, and clay size; statements 295 through 350.
- (10) Calculate and punch mean, standard deviation, skewness, and kurtosis; statements 355 through 530.
- (11) If the amount of each coarse-fraction constituent in the total sample has not been calculated, go to (12); if it has, go to (15); statement 535.
- (12) Calculate and punch fraction of total sample made up of each coarse fraction component; statements 540 through 745.
- (13) Subtract from each size class the amount of nonresistant material present; statement 680.
- (14) Return to (8), repeat program through (11). All calculations this time use only the resistant minerals, statement 750.
- (15) Is there data for another sample? If so, go to (3); if not, go to (16); statement 755.
- (16) End; statement 760.

Table 1. - Listing of program for standard-size analysis of unconsolidated sediments.

```

C THIS PROGRAM IS DESIGNED TO COMPUTE MEAN, STANDARD DEVIATION,
C SKEWNESS, KURTOSIS, SAND PERCENT, SILT PERCENT, CLAY PERCENT, AND
C PERCENT OF COARSE FRACTIONS OF SEDIMENT DATA USING PHI SIZES VS
C WEIGHT PERCENT.
C PROGRAMMED IN FORTRAN II FOR IBM 1620 BY J. PIERCE AND D. GOOD
C
5 DIMENSION FRGRS (22), BEAKR (22), PIP (8), AMPHI (22),
1FRNT (22), WTPIP (8), FRPCT (22), FPPCT (8), PMPHI(8), QTZ (22),
2YQTZ(22), OPAQ(22), SHELL(22), ORG(22), CAGG(22), FAGG(22),
3PLFRG(22), FMUD(22), ECH(22)
15 K = 0
20 READ 25, N
25 FORMAT (I5)
30 K = K+ 1
C
35 DO 54 I = 1,22,1
40 FRGRS (I) = 0.0
45 BEAKR (I) = 0.0
50 AMPHI (I) = 0.0
51 FRNT (I) = 0.0
FRPCT(I) = 0.0
QTZ (I) = 0.0
YQTZ (I) = 0.0
OPAQ (I) = 0.0
SHELL (I) = 0.0
ORG (I) = 0.0
CAGG (I) = 0.0
FAGG (I) = 0.0
PLFRG (I) = 0.0
FMUD (I) = 0.0

```

```

54 ECH (I) = 0.0
C
55 DO 63 J = 1,8,1
60 PIP (J) = 0.0
61 PMPHI (J) = 0.0
62 WTPIP (J) = 0.0
63 FPPCT (J) = 0.0
C
65 READ 70, GROSS, BEAK, M
70 FORMAT (2F10.3, I5)
C
74 DO 75 I = 1, M, 1
75 READ 80, FRGRS(I), BEAKR(I), AMPHI(I)
80 FORMAT (2F10.3, F7.2)
81 READ 82, P
82 FORMAT (F1.0)
83 IF (P - 1.) 94, 85, 94
85 READ 90, (PIP(J), PMPHI(J), J = 1,8,1)
90 FORMAT (4(F10.4, F8.2))
94 DO 95 I = 1, M, 1
95 READ 100, QTZ (I), YQTZ (I), OPAQ(I), SHELL(I), ORG(I), CAGG(I),
1FAGG(I), PLFRG(I), FMUD(I), ECH(I)
100 FORMAT (10F5.2)
C
105 LL = 0
110 SMPNT = GROSS - BEAK
120 DO 125 I = 1, M, 1
125 FRNT (I) = FRGRS (I) - BEAKR (I)
C
130 LL = LL + 1
C
135 SUMFR = 0.0
140 DO 145 I = 1, M, 1
145 SUMFR = SUMFR + FRNT (I)
C
150 IF (LL - 2 ) 155, 170, 170
155 SLOSS = SMPNT - SUMFR
156 PUNCH 157
157 FORMAT (30H      SUMFR      LOSS      K      LL  )
160 PUNCH 165, SUMFR, SLOSS, K, LL
165 FORMAT (2F10.3, 2I5)
C
170 IF (P - 1.) 250, 174, 250
174 WTPIP(8) = 50.0 * (PIP(8) - 0.0066)
175 J = 8
180 J = J - 1
185 L = J + 1
190 WTPIP (J) = 50.0*(PIP (J) - PIP (L) )
195 IF (WTPIP (J)) 200, 205, 205
200 WTPIP (J) = 0.0
205 IF ( J - 1) 210, 210, 180
C
210 IF (SENSE SWITCH 1) 211, 225
211 PUNCH 212
212 FORMAT (/17H      WTPIP(J)      J      )
215 PUNCH 220, (WTPIP(J), J, J = 1,8,1)
220 FORMAT (5(F12.4, I4))
C
225 SUMWP = 0.0
230 DO 235 J = 1,8,1

```

```

235 SUMWP = SUMWP + WTPIP (J)
C
240 SUMFR1 = SUMFR - FRNT (M) + SUMWP
245 SUMFR = SUMFR1
246 FRNT (M) = SUMWP
C
250 DIV = 1.0 / SUMFR
251 PUNCH 252
252 FORMAT (/30HFRPCT(I)   K   LL   I   4 ACROSS   )
255 DO 260 I = 1,M,1
260 FRPCT (I) = FRNT (I)*DIV
265 PUNCH 270, (FRPCT(I), K, LL, I, I = 1,M,1)
270 FORMAT (4(F8.4, 3I4))
275 IF (P - 1.) 295, 276, 295
C
276 PUNCH 277
277 FORMAT (/35HFPPCT(J)   K   LL   J   4 ACROSS   )
C
280 DO 285 J = 1,8,1
285 FPPCT (J) = WTPIP (J)*DIV
290 PUNCH 270, (FPPCT(J), K, LL, J, J = 1,8,1)
C
295 SDPCT = 0.0
300 IM = M - 1
305 DO 310 I = 1, IM, 1
310 SDPCT = SDPCT + FRPCT (I)
311 IF (P - 1.) 343, 315, 343
C
315 SLPCT = 0.0
320 DO 325 J = 1,6,1
325 SLPCT = SLPCT + FPPCT (J)
C
330 CLPCT = 0.0
335 DO 340 J = 7,8,1
340 CLPCT = CLPCT + FPPCT (J)
GO TO 341
343 SLPCT = 0.0
CLPCT = 0.0
C
341 PUNCH 342
342 FORMAT (/40H           SDPCT   SLPCT   CLPCT   K   LL   )
345 PUNCH 350, SDPCT, SLPCT, CLPCT, K, LL
350 FORMAT (3F10.4, 2I5)
C
355 SUMX1 = 0.0
360 SUMX2 = 0.0
361 IF ( P- 1.0 ) 364 , 362, 364
362 MM = IM
363 GO TO 365
364 MM = M
365 DO 370 I = 1,MM,1
370 SUMX1 = SUMX1 + FRNT (I)*AMPHI (I)
C
375 IF (P - 1.) 390, 380, 390
380 DO 385 J = 1,8,1
385 SUMX2 = SUMX2 + PMPHI (J)*WTPIP (J)
390 SUMX = SUMX1 + SUMX2
C
395 XMEAN = SUMX*DIV
C
400 SUM2A = 0.0

```

```

405 SUM3A = 0.0
410 SUM4A = 0.0
415 DO 435 I= 1, MM, 1
420 SUM2A = SUM2A+ FRNT(I)*(AMPHI(I)-XMEAN)*(AMPHI(I)-XMEAN)
425 SUM3A = SUM3A+FRNT(I)*(AMPHI(I)-XMEAN)*(AMPHI(I)-XMEAN)*(AMPHI(I)
1 -XMEAN)
435 SUM4A = SUM4A+FRNT(I)*(AMPHI(I)-XMEAN)*(AMPHI(I)-XMEAN)*(AMPHI(I)-
1XMEAN)*(AMPHI(I)-XMEAN)
C
440 SUM2B = 0.0
445 SUM3B = 0.0
450 SUM4B = 0.0
C
455 IF (P - 1.) 485, 460, 485
460 DO 480 J = 1,8,1
465 AB = PMPHI(J) - XMEAN
470 SUM2B = SUM2B + WTPIP (J)*AB*AB
475 SUM3B = SUM3B + WTPIP (J)*AB*AB*AB
480 SUM4B = SUM4B + WTPIP(J)*AB*AB*AB*AB
C
485 SUM2 = SUM2A + SUM2B
490 SUM3 = SUM3A + SUM3B
495 SUM4 = SUM4A + SUM4B
C
500 SIGMA = SQRTF (SUM2*DIV)
505 SIGMA3 = SIGMA*SIGMA*SIGMA
510 SIGMA4 = SIGMA*SIGMA*SIGMA*SIGMA
C
515 SKEW = (0.5*SUM3*DIV)/SIGMA3
520 CURT = 0.5*(((SUM4*DIV)/SIGMA4) - 3.0)
521 PUNCH 522
522 FORMAT (/50H      XMEAN      SIGMA      SKEW      CURT      K      LL  )
525 PUNCH 530, XMEAN, SIGMA, SKEW, CURT, K, LL
530 FORMAT (4F10.3, 2I5)
C
535 IF (LL-2) 540,755,755
540 SQTZ = 0.0
545 SYQTZ = 0.0
550 SSHELL = 0.0
555 SOPAQ = 0.0
560 SORG = 0.0
565 SCAGG = 0.0
570 SFAGG = 0.0
575 SPLFRG = 0.0
580 SFMUD = 0.0
585 SECH = 0.0
C
590 DO 680 I = 1,M,1
595 SQTZ = SQTZ + QTZ (I)*FRNT(I)
600 SYQTZ = SYQTZ + YQTZ (I)*FRNT(I)
605 SOPAQ = SOPAQ + OPAQ(I)*FRNT(I)
610 A1 = SHELL(I)*FRNT(I)
615 SSHELL = SSHELL + A1
620 A2 = ORG(I)*FRNT(I)
625 SORG = SORG + A2
630 A3 = CAGG(I)*FRNT(I)
635 SCAGG = SCAGG + A3
640 A4 = FAGG(I)*FRNT(I)
645 SFAGG = SFAGG + A4
650 A5 = PLFRG(I)*FRNT(I)
655 SPLFRG = SPLFRG + A5

```

```

660 A6 = FMUD(I) * FRNT(I)
665 SFMUD = SFMUD + A6
670 A7 = ECH(I)*FRNT(I)
675 SECH = SECH + A7
680 FRNT (I) = FRNT(I) -A1-A2-A3-A4-A5-A6-A7
C
685 PQTZ = SQTZ*DIV
690 PYQTZ = SYQTZ*DIV
700 PSHELL = SSHELL*DIV
705 POPAQ = SOPAQ*DIV
710 PORG = SORG*DIV
715 PCAGG = SCAGG*DIV
720 PFAGG = SFAGG*DIV
725 PPLFRG = SPLFRG*DIV
730 PFMUD = SFMUD * DIV
735 PECH = SECH*DIV
736 PUNCH 737
737 FORMAT(/80H  PQTZ  PYQTZ  POPAQ  PSHELL  PORG  PCAGG  PFAGG  PPLFR
1G  PFMUD  PECH  K  LL)
740 PUNCH 745, PQTZ, PYQTZ, POPAQ, PSHELL, PORG, PCAGG, PFAGG,
1PPLFRG,PFMUD, PECH, K, LL
745 FORMAT (10F7.4,2I5)
C
750 IF (LL-2) 130,755,755
C
755 IF (K-N) 30, 756,756
756 TYPE 757
757 FORMAT (7HTHE END)
758 GO TO 15
C
760 END

```

INPUT TO SIZE-ANALYSIS PROGRAM

The specific form of data input for each sample may range within certain specified limits. These limits are explained in the input form below and are allowed by control statements in the program. A listing of input for two sample problems is shown in Table 2. The first data card immediately follows the last card of the object deck.

- | | |
|-----------------------|---|
| First header card. | Value of N, the number of samples to be processed. This is an integer with the last digit in column 5. |
| Second header card. | Values for gross sample weight, tare weight of weighing vessel, and number of size classes greater than 62 microns. Gross sample weight, GROSS, to three decimal places in columns 1-10 with the decimal point in column 7; tare weight, BEAK, to 3 decimal places, in columns 11-20 with the decimal point in column 17; number of size classes, M, an integer, in columns 21-25 with the last digit in column 25. |
| First set of M cards. | Values of FRGRS(I), gross weight of a size class; BEAKR(I), weight of weighing vessel; and AMPHI(I), midpoint of size range, in phi units. One card is made up for each size class. There will be as many cards as there are size classes in the greater-than-62-micron size. The number of cards must agree with M, on the previous card. FRGRS(I), to three decimal places, is in columns 1-10 with the decimal point in column 7; BEAKR(I), to three decimal places, in columns 11-20 with the decimal point in column 17; AMPHI(I), with two decimal places, in columns 21-27, with the decimal point in column 25. |

Third header card.	Value of P. This card signifies presence or absence of pipette data. No pipette analysis is indicated by 0 (zero) in column 1; pipette analysis, by 1 in column 1.
First card with pipette analysis data.	Values of net weight of sediment and dispersant recovered in the first four aliquot portions of pipette analysis, and the midpoint of each size class, in phi units. If no pipette data available, as signified by a 0 (zero) on the previous card, this and the following card will not be in the data deck. The first value on the card is the net weight of the first aliquot; the second value, the midpoint of this size class; the third value, the net weight of the second size class, etc. The net weight of the sediment has the form of xx.xxxx and the midpoint, x.xx. The decimal points are located in columns 6, 16, 24, 34, 42, 52, 60, and 70.
Second card with pipette analysis data.	Value of net weight of sediment plus dispersant for the second four aliquots of the pipette analysis. These values have the same format as the previous card.
Second set of M cards.	Decimal fraction of each size class composed of 10 coarse fraction components. There is one card for each size class and the number of cards must agree with M from the second header card. These numbers are of the form oo.xx with the decimal points located in columns 3, 8, 13, 18, 23, 28, 33, 38, 43, and 48 respectively.

INPUT FOR SAMPLE PROBLEMS

Data for two sample problems are listed in Table 2. The two problems will give some idea of the variability allowed in the data input. Each line of the listing is one punch card.

The first data card indicates the number of samples to be processed, in this case, two. The data obtained by the analyses immediately follow this card.

The first data card for each sample contains the gross weight of sample plus weighing vessel, weight of vessel, and number of sieve classes in the sample. These are shown on lines 2 and 48 (Table 2).

The next set of cards, equal in number to the number of classes shown on the first data card, contain the gross weight of the size fraction plus weighing vessel, weight of the vessel, and the midpoint, in phi-scale units, of the size class. For the first sample, there are 22 cards (lines 3 through 24); for the second, 13 (lines 49 through 61). Note that, for sample 2, on line 61, the pan fraction will have a net weight of 0.184 grams. The sum of the pipette weights will be substituted for this value and it will not enter the computations in the program. This illustrates that where a pipette analysis is present, as in the case of the second sample, the wet-sieve separation of the mud and sand must be thorough.

The next card, on line 25 for sample 1 and line 62 for sample 2, is a control card indicating whether a pipette analysis was made. No pipette data are present for the first sample, as indicated by a zero (0) in column 1. The numeral 1 in column 1 on line 62 indicates that a pipette analysis was performed on the second sample.

If a pipette analysis was made and so indicated on the control card, the next two data cards contain the results of the analysis as shown for the second sample, lines 63 and 64. Four weights and associated midpoint of the size class are punched on each card. The program allows for a maximum of eight size classes to be differentiated by pipette analysis.

These two cards are absent from the data deck if the control card so indicates, as shown for the first sample. No pipette data are present for this sample.

The next set of cards, equal in number to the number of sieve classes, contains the decimal fraction estimates of 10 components in the sieve classes. There are 22 cards for the first sample (lines 26 through 47); 13, for the second (lines 65 through 77). Although space is allotted for 10 different categories, only 4, at most, have been used for each size class. The last card of the series for both samples, indicative of the material caught in the pan, shows that it is all mud (within silt or clay sizes).

Table 2. - Input for sample problems.

2				1
102.299	50.000	22		2
28.956	28.674	-1.25		3
28.899	28.662	-0.87		4
32.447	31.952	-0.62		5
29.542	28.802	-0.37		6
30.187	28.902	-0.12		7
33.856	31.954	0.12		8
31.812	28.820	0.37		9
36.874	32.071	0.62		10
36.965	28.647	0.87		11
39.475	29.065	1.12		12
50.973	43.209	1.37		13
47.960	42.617	1.62		14
46.010	42.840	1.87		15
44.272	42.789	2.12		16
43.408	42.783	2.37		17
43.318	43.003	2.62		18
42.992	42.901	2.87		19
42.798	42.770	3.12		20
43.270	43.260	3.37		21
42.859	42.855	3.62		22
42.960	42.957	3.87		23
28.992	28.888	4.25		24
0				25
1.00				26
.98	.01	.01		27
.98	.01	.01		28
.98	.01	.01		29
.98	.01	.01		30
.97	.02	.01		31
.97	.02	.01		32
.97	.02	.01		33
.97	.02	.01		34

.96	.01	.01	.02							35
.96	.01	.01	.02							36
.96	.01	.01	.02							37
.96	.01	.01	.02							38
.96	.01	.01	.02							39
.96	.01	.01	.02							40
.96	.01	.01	.02							41
.96	.01	.01	.02							42
.96	.01	.01	.02							43
.96	.01	.01	.02							44
.96	.01	.01	.02							45
.96	.01	.01	.02							46
				1.00						47
70.796	30.312	13								48
28.718	28.674	-0.25								49
28.709	28.662	0.25								50
32.097	31.952	0.75								51
29.594	28.902	1.25								52
32.915	31.954	1.62								53
31.456	28.820	1.87								54
35.681	32.071	2.12								55
38.137	28.647	2.37								56
43.959	29.065	2.62								57
48.316	43.209	2.87								58
44.589	42.617	3.25								59
43.347	42.840	3.75								60
42.973	42.789	4.25								61
1										62
.4292	4.25	.3486	4.75	.3114	5.25	.2745	5.75			63
.2438	6.50	.2024	7.50	.1697	8.50	.1421	9.50			64
.83	.10	.05	.02							65
.83	.10	.05	.02							66
.83	.10	.05	.02							67
.95		.05								68
.95		.05								69
.95		.05								70
.95	.02	.01	.02							71
.95	.02	.01	.02							72
.95	.02	.01	.02							73
.95	.02	.01	.02							74
.76	.22	.01	.01							75
.76	.22	.01	.01							76
				1.00						77

FORM OF OUTPUT

Output for this size-analysis program is in the form of punch cards. A tabulation of the output associated with the two sample problems is given in Table 3.

Each group of output variables is preceded by a heading card which, in most cases, is punched with a symbolic abbreviation of the variable. This abbreviation should immediately suggest the implied meaning. Blank cards, interspersed throughout the deck, as part of the output, provide spaces in the tabulation to increase legibility.

The maximum number of cards to comprise the output for one sample is 41. Of this number, 10 are heading cards, 9 are blank spacers, and 22 contain the calculated results.

The first output card is a header card containing the names of two variables: SUMFR and LOSS. Counting indices, K and LL, are also punched on this card. The second card contains the values of the above variables and indices. The variable SUMFR is the sum of net weights of various size fractions; and LOSS is the difference between the sum of fraction weights and original weight of sample. This value is indicative of the amount of material lost during analysis. The index, K, indicates the number of the sample being processed and is consecutive from one. The index, LL, indicates whether output is based on all the material in the sample (LL = 1) or only the material that is resistant to weathering (LL = 2). Only two states, 1 or 2, are possible for LL. A spacer card occurs after the first two cards.

The next group of cards contains the fraction of the total sample in each sieve-size class. These are preceded by a header card containing the names FRPCT(I), K, LL, I, 4 ACROSS. FRPCT(I) is the fraction of the total in a size class, K and LL are explained above, and I is an index, numbered consecutively from 1 through M, or number of sieve classes, which ties the fraction value to the size class. Size class 1 contains the largest particles. The words "4 ACROSS" indicate there are 4 values each of FRPCT(I), K, LL, and I on one punch card.

If a pipette analysis was performed, the next three cards, exclusive of spacers, contain the calculated results of this analysis and an associated header card. FPPCT(J) is the fraction of the total sample in each pipette class. J is an index, numbered consecutively from 1 through 8, indicating the pipette class. Again, as with the sieve classes, values for four size classes are punched on each card. If a pipette analysis was not performed, these cards are absent.

The next two cards are a heading card and the fraction of total sample made up of sand and larger size particles, of silt, and of clay. If no pipette analysis was performed, no differentiation between the last two can be made and a value for neither is punched. The fraction of the total sample in the fine sizes can be obtained by subtracting the sand fraction from unity.

The next two cards deal with the mean, standard deviation (SIGMA), skewness, and kurtosis values for all the material.

Following these two cards, are a header card and a card containing values associated with 10 named variables. These values are the fraction of the sand-size material made up of 10 different components. The components used in this study are discussed in the section on laboratory techniques. Components used are immaterial to the computational procedures of the program.

The remaining cards, equal in number to the previous ones, less 2, are results of calculations involving only the size distribution of the resistant minerals. The estimated amount of all components, except for quartz and opaque minerals, have been subtracted from the total. When a pipette analysis is used the distribution of the silt- and clay-size material is considered.

OUTPUT FOR SAMPLE PROBLEMS

Output for two samples is given in Table 3. There are 70 cards, including blank cards and headings, in this output deck.

The sum of the net weight of the size fractions, and analytical loss are shown on line 2 for sample 1 and line 35 for sample 2. For the first sample, line 2, the sum of the net weights is 50.404 grams. This value is 1.895 grams less than the original weight of the sample. The sum is 40.289 grams for the second sample (line 35). A loss of 0.195 grams occurred during analysis. This value can be either negative or positive. For the negative case, the sum would be more than the original sample weight, probably due to recording or weighing errors. A positive value can be ascribed to weighing error, recording mistakes, or an extremely large analytical loss. Notice that K equals 1 for the first sample and 2 for the second.

The fraction of the total sample in each sieve-size class is tabulated on lines 5 through 10 for the first sample and 38 through 41 for the second. The coarsest sieve size is 0.55 percent of the total sample in the first sample; the second coarsest, 0.47 percent; and the finest, or pan fraction, 0.20 percent.

For the second sample, the two coarsest grades of material each contain 0.07 percent of the total. The finest size class in the second sample contains, according to the output, 34.50 percent of the total sample. The sum of the pipette fractions is 34.48 percent. The sum of the pipette fractions have been substituted for the pan fraction, and the discrepancy, 0.02 percent, is due to truncation of values to four significant figures during output operation.

Fractions of the total sample in each of eight pipette classes are shown on lines 44 and 45. The coarse silt size accounts for 6.58 percent of the total sample, the less-than-nine-phi class, 11.06 percent. Because no pipette analysis was performed on the first sample, punch cards giving the pipette results are absent from the output deck.

The sand, silt, and clay fractions of the total sample are shown on line 13 for sample 1 and line 48 for sample 2. The first sample is 99.79 percent sand. No figures are given for the amount of silt and clay because no differentiation was made. The second sample is 65.49 percent sand, 21.19 percent silt, and 13.31 percent clay. In all cases, the sum of the sand, silt, and clay fractions are less than unity, which is due to truncation of values during output.

Table 3. - Output for sample problems.

SUMFR	LOSS	K	LL												
50.404	1.895	1	1												
FRPCT(I)	K	LL	I	4	ACROSS										
.0055	1	1	1	.0047	1	1	2	.0098	1	1	3	.0146	1	1	4
.0254	1	1	5	.0377	1	1	6	.0593	1	1	7	.0952	1	1	8
.1650	1	1	9	.2065	1	1	10	.1540	1	1	11	.1060	1	1	12
.0628	1	1	13	.0294	1	1	14	.0123	1	1	15	.0062	1	1	16
.0018	1	1	17	.0005	1	1	18	.0001	1	1	19	0.0000	1	1	20
0.0000	1	1	21	.0020	1	1	22								10
SDPCT	SLPCT	CLPCT	K	LL											
.9979	0.0000	0.0000	1	1											
XMEAN	SIGMA	SKEW	CURT	K	LL										
1.060	.646	-.117	.983	1	1										
PQTZ	PYQTZ	POPAQ	PSHELL	PORG	PCAGG	PFAGG	PPLFRG	PFMUD	PECH	K	LL				
.9629	.0134	0.0000	.0099	.0116	0.0000	0.0000	0.0000	.0020	0.0000	1	1				
FRPCT(I)	K	LL	I	4	ACROSS										
.0057	1	2	1	.0047	1	2	2	.0099	1	2	3	.0148	1	2	4
.0258	1	2	5	.0382	1	2	6	.0601	1	2	7	.0966	1	2	8
.1673	1	2	9	.2051	1	2	10	.1530	1	2	11	.1053	1	2	12
.0624	1	2	13	.0292	1	2	14	.0123	1	2	15	.0062	1	2	16
.0017	1	2	17	.0005	1	2	18	.0001	1	2	19	0.0000	1	2	20
0.0000	1	2	21	0.0000	1	2	22								
SDPCT	SLPCT	CLPCT	K	LL											
.9999	0.0000	0.0000	1	2											30
XMEAN	SIGMA	SKEW	CURT	K	LL										
1.048	.631	-.243	.553	1	2										
SUMFR	LOSS	K	LL												
40.289	.195	2	1												
FRPCT(I)	K	LL	I	4	ACROSS										
.0007	2	1	1	.0007	2	1	2	.0023	2	1	3	.0113	2	1	4
.0156	2	1	5	.0430	2	1	6	.0589	2	1	7	.1549	2	1	8
.2432	2	1	9	.0834	2	1	10	.0322	2	1	11	.0082	2	1	12
.3450	2	1	13												
FPPCT(J)	K	LL	J	4	ACROSS										
.0658	2	1	1	.0303	2	1	2	.0301	2	1	3	.0250	2	1	4
.0338	2	1	5	.0267	2	1	6	.0225	2	1	7	.1106	2	1	8
SDPCT	SLPCT	CLPCT	K	LL											
.6549	.2119	.1331	2	1											
XMEAN	SIGMA	SKEW	CURT	K	LL										
4.015	2.472	.634	.112	2	1										
PQTZ	PYQTZ	POPAQ	PSHELL	PORG	PCAGG	PFAGG	PPLFRG	PFMUD	PECH	K	LL				
.6140	0.0000	.0197	.0061	.0149	0.0000	0.0000	0.0000	.3450	0.0000	2	1				
FRPCT(I)	K	LL	I	4	ACROSS										
.0006	2	2	1	.0006	2	2	2	.0020	2	2	3	.0109	2	2	4

•0152	2	2	5	•0417	2	2	6	•0584	2	2	7	•1535	2	2	8
•2410	2	2	9	•0826	2	2	10	•0322	2	2	11	•0082	2	2	12
•3525	2	2	13												60
FPPCT(J)	K	LL	J	4	ACROSS										
•0672	2	2	1	•0310	2	2	2	•0307	2	2	3	•0256	2	2	4
•0345	2	2	5	•0272	2	2	6	•0230	2	2	7	•1130	2	2	8
SDPCT		SLPCT		CLPCT	K	LL									
•6474		•2165		•1360	2	2									
XMEAN		SIGMA		SKEW	CURT	K	LL								
4.051		2.485		•620	•071	2	2								70

The parameters, mean, standard deviation, skewness, and kurtosis, for size distribution of all material in the sample, are given on lines 16 and 51.

Results of "coarse-fraction" analysis are given on lines 19 and 54. Each value is the fraction of the sand-size material composed of one of 10 different components. In the first sample, 5 components each made up less than 0.01 percent of the total; in the second, 5 components again were present in negligible amounts.

The remaining output data pertains to the distribution of more resistant minerals. This is indicated in the output by setting the counter, LL, equal to two. In essence, the output format is repeated for the fractions in each size class, including sizes in the pipette range, the sand, silt, and clay fractions, and the four size parameters. For those samples with sieve analysis only, the amount of material in the less-than-sand size is not considered in the calculations. The values associated with pipette analysis are used in the calculations for those samples on which such analyses were performed.

REFERENCES

- Inman, D. L., 1952, Measures for describing the size distribution of sediments: *Jour. Sed. Pet.*, v. 22, no. 3, p. 125-145.
- Krumbein, W. C., and Pettijohn, F. J., 1938, *Manual of sedimentary petrography*: Appleton-Century-Crofts, New York, 549 p.
- Mason, C. C., and Folk, R. L., 1958, Differentiation of beach, dune, and aeolian flat environments by size analysis, Mustang Island, Texas: *Jour. Sed. Pet.*, v. 28, no. 2, p. 211-226.
- Milner, H. B., 1962, *Sedimentary petrography*: Macmillan, New York, 2 vols, 715 p.
- Shepard, F. P., 1963, *Submarine geology*, 2nd edition: Harper and Row, New York, 557 p.
- Shepard, F. P., and Moore, D. G., 1954, Sedimentary environments differentiated by coarse-fraction studies: *Am. Assoc. Petroleum Geologists Bull.*, v. 38, no. 8, p. 1792-1802.
- Trask, P. D., 1932, *Origin and environment of source sediments of petroleum*: Gulf Publ. Co., Houston, 323 p.

PROGRAM ABSTRACT

Title (If subroutine state in title):

FORTRAN II Program for Standard-Size Analysis of Unconsolidated Sediments

Computer: IBM 1620

Date: January 1965

Programming language: FORTRAN II

Author, organization: Jack W. Pierce, U.S. National Museum, Division of Sedimentology, and George Washington Univ., Washington, D.C.; or Donald I. Good, Kansas Geological Survey, Lawrence, Kansas.

Direct inquiries to: Authors, or

Name: Daniel F. Merriam

Address: Kansas Geological Survey

Lawrence, Kansas

Purpose/description: Calculate the mean, standard deviation, skewness, and kurtosis for standard sediment-size analysis; the fraction of the total sample in each size class; the fraction of sand-size material composed of up to 10 different components

Mathematical method: Statistical method of moments for size parameters.

Restrictions, range: _____

Storage requirements: _____

Equipment specifications:

Memory 20K _____ 40K _____ 60K _____ K _____

Automatic divide: Yes No _____ Indirect addressing: Yes

Other special features required _____

Additional remarks (include at author's discretion: fixed/float, relocatability; optional: running time, number of times run successfully, programming hours) Less than 5 hours were needed to process 400 samples.

COMPUTER CONTRIBUTIONS
 Kansas Geological Survey
 University of Kansas
 Lawrence, Kansas

Special Distribution Publication

3.	BALGOL program for trend-surface mapping using an IBM 7090 computer, by J. W. Harbaugh, 1963	\$ 0.50
4.	FORTTRAN II program for coefficient of association (Match-Coeff) using an IBM 1620 computer, by R. L. Kaesler, F. W. Preston, and D. I. Good, 1963	\$ 0.25
9.	BALGOL programs for calculation of distance coefficients and correlation coefficients using an IBM 7090 computer, by J. W. Harbaugh, 1964	\$ 0.75
11.	Trend-surface analysis of regional and residual components of geologic structure in Kansas, by D. F. Merriam and J. W. Harbaugh, 1964	\$ 0.75
12.	FORTTRAN and FAP program for calculating and plotting time-trend curves using an IBM 7090 or 7094/1401 computer system, by W. T. Fox, 1964	\$ 0.75
13.	FORTTRAN program for factor and vector analysis of geologic data using an IBM 7090 or 7094/1401 computer system, by Vincent Manson and John Imbrie, 1964	\$ 1.00
14.	FORTTRAN II trend-surface program for the IBM 1620, by D. I. Good, 1964	\$ 1.00
15.	Application of factor analysis to petrologic variations of Americus Limestone (Lower Permian), Kansas and Oklahoma, by J. W. Harbaugh and Ferruh Demirmen, 1964	\$ 1.00
23.	ALGOL program for cross-association of nonnumeric sequences using a medium-size computer, by M. J. Sackin, P. H. A. Sneath, and D. F. Merriam, 1965	\$ 0.75
24.	BALGOL program and geologic application for single and double Fourier series using IBM 7090/7094 computers, by F. W. Preston and J. W. Harbaugh, 1965	\$ 1.00
26.	FORTTRAN II trend-surface program with unrestricted input for the IBM 1620 computer, by R. J. Sampson and J. C. Davis, 1966	\$ 0.50
27.	Application of factor analysis to a facies study of the Leavenworth Limestone (Pennsylvanian-Virgilian) of Kansas and Environs, by D. F. Toomey, 1966	\$ 0.75
28.	FORTTRAN II program for standard-size analysis of unconsolidated sediments, by J. W. Pierce and D. I. Good, 1966	\$ 0.75

Report of Studies

170-3	Mathematical conversion of section, township, and range notation to Cartesian Coordinates, by D. I. Good, 1964	\$ 0.50
-------	--	---------

Bulletin

171	A computer method for four-variable trend analysis illustrated by a study of oil-gravity variations in southeastern Kansas, by J. W. Harbaugh, 1964	\$ 1.00
-----	---	---------

Reprints (available for limited time)

Use of asymmetric frequency distribution curves of core analysis data in calculating oil reserves, by F. W. Preston and J. S. Van Scoyoc (reprinted from <u>Stanford University Publications</u> , Geological Sciences, v. 9, no. 2, 1964)	no charge
Pattern recognition studies of geologic structure using trend-surface analysis, by D. F. Merriam and R. H. Lippert (reprinted from <u>Colorado School Mines Quarterly</u> , v. 59, no. 4, 1964)	no charge
Trend-surface mapping of hydrodynamic oil traps with the IBM 7090/7094 computer, by J. W. Harbaugh (reprinted from <u>Colorado School Mines Quarterly</u> , v. 59, no. 4, 1964)	no charge
Finding the ideal cyclothem, by W. C. Pearn (reprinted from Symposium on cyclic sedimentation, D. F. Merriam, editor, <u>Kansas Geological Survey Bulletin</u> 169, v. 2, 1964)	no charge
Fourier series characterization of cyclic sediments for stratigraphic correlation, by F. W. Preston and J. H. Henderson (reprinted from Symposium on cyclic sedimentation, D. F. Merriam, editor, <u>Kansas Geological Survey Bulletin</u> 169, v. 2, 1964)	no charge
Fourier series analysis in geology, by J. W. Harbaugh and F. W. Preston (reprinted from <u>College of Mines, Arizona University</u> , v. 1, 1965)	no charge
Geology and the computer, by D. F. Merriam (reprinted from <u>New Scientist</u> , v. 26, no. 444, 1965)	no charge
Application of factor analysis to the Upper Cambrian Reagan Sandstone of central and northwest Kansas, by M. N. McElroy and R. L. Kaesler (reprinted from <u>The Compass</u> , v. 42, no. 3, 1965)	no charge
Quantitative comparison of contour maps, by D. F. Merriam and P. H. A. Sneath (reprinted from <u>Journal of Geophysical Research</u> , v. 71, no. 4, 1966)	no charge
Trend-surface analysis of stratigraphic thickness data from some Namurian rocks east of Sterling, Scotland, by W. A. Read and D. F. Merriam (reprinted from <u>Scottish Journal of Geology</u> , v. 2, pt. 1, 1966)	no charge

